
Corso di Progettazione Assistita delle Strutture meccaniche 
Esercitazione sul submodelling 

 
La fig.1 mostra un tipico albero rotante, soggetto ad una forza verticale F applicata sulla mezzeria, e 
semplicemente appoggiato agli estremi; questa esercitazione ha lo scopo di stimare il campo di tensioni che 
si sviluppa in prossimità dello spallamento (si fa notare come il livello tensionale effettivo sia sensibilmente 
maggiore rispetto a quello nominale, a causa della brusca variazione di sezione). 

Fig.1. Schema di carico e dimensioni di un albero rotante. 

 
 

Il calcolo può essere condotto in due modi: 
 
(A) Calcolo manuale 
 
Secondo la teoria delle travi di De Saint Venant, la tensione nominale massima in direzione assiale si trova 
sul bordo, e vale: 

dove Mf e dp sono rispettivamente il momento flettente ed il diametro relativi alla sezione interessata. 
La tensione locale effettiva σ è invece pari al prodotto fra la σn ed il coefficiente di intaglio teorico Kt, il cui 
valore è ricavabile da opportuni grafici in funzione del raggio di raccordo dell’intaglio, delle dimensioni e 
della forma della sezione, e del modo di carico (in questo caso 2≈tK ): 

Sfruttando i dati di Fig.1 si ottiene dunque: 
 
σ≈30 N/mm2 
 
(B) Calcolo con gli elementi finiti 
 
La validazione del calcolo manuale può essere condotta sfruttando il metodo degli elementi finiti, ed in 
particolare la tecnica del submodelling: inizialmente viene sviluppato un modello dell’albero in cui è assente 



il raccordo, dopodiché viene realizzato un sottomodello, contenente il raccordo, ed utilizzato per studiare le 
sollecitazioni che si sviluppano in questo. 
Il modello dell’albero privo di raccordo è chiamato coarse, mentre il sottomodello è chiamato sottomodello 
(appunto). I passi necessari all’analisi di un sottomodello sono i seguenti: 

1. creare, salvare ed analizzare il modello grossolano (coarse) 
2. svuotare il database e cambiare il jobname (/filname,nuovo_nome,db)per evitare di sovrascrivere i 

file generati nel punto 1. 
3. creare il sottomodello, utilizzando lo stesso tipo di elementi, le stesse real constant e gli stessi 

materiali (anche se indicati con numeri diversi). Importante: la posizione del sottomodello rispetto 
al riferimento cartesiano globale deve corrispondere alla porzione del modello coarse che si 
vuole studiare. 

4. effettuare l’interpolazione degli spostamenti sui bordi di taglio del sottomodello; si articola nei 
seguenti passi: (a) selezionare i nodi sui bordi di taglio del sottomodello (nsel) e salvarli in un file 
(default jobname.node) con il comando nwrite. (b) riselezionare tutte le entità (all) uscire dal 
prepocessore e salvare il sottomodello. (c) richiamare nel database il modello grossolano 
(resume,coarse,db), entrare nel postprocessore (/post1), indicare da quale file si vogliono leggere i 
risultati (file,coarse,rst) e leggere i risultati dell’analisi sul modello grossolano (set,1,1). (d) 
effettuare l’interpolazione con il comando cbdof, il quale genera un file di testo chiamato, di default, 
jobname.cbdo. (e) alla fine si può lasciare il postprocessore, e ricaricare nel database il sottomodello: 
resume,sottomodello,db. 

5. si entra nell’ambiente di soluzione (/solu) e si impongono gli spostamenti interpolati attraverso la 
lettura del file .cbdo prima citato: /input,jobname,cbdo. Si devono duplicare, sul sottomodello, tutti 
i carichi o le condizioni al contorno che esistevano sul modello grossolano. 

6. si può analizzare il sottomodello (solve) e postprocessare i risultati.  
7. si deve necessariamente controllare che le tensioni all’interfaccia fra il modello corse ed il 

sottomodello siano ragionevolmente vicine; se così non è l’analisi con il sottomodello non può 
essere considerata valida. 

 
Qui di seguito viene riportato uno dei possibili listati di comandi ANSYS idoneo allo scopo 
dell’esercitazione: 
 
finish 
/clear 
/filname,coarse 
c*** studio del raccordo di un albero attraverso l'uso di un sottomodello 
c*** modello 'coarse' (grossolano): privo di raccordo tra perno e albero 
c***macro contenente i parametri di modellazione 
*create,ass_par 
dp=30 !!diametro di perno del cuscinetto 
da=40 !!diametro dell'albero 
bc=20 !!larghezza di fascia del cuscinetto 
Lp=30 !!lunghezza del perno 
La=200 !!lunghezza dell'albero 
r=2 !!2r<(da-dp)/2; raggio di raccordo 
c***elementi e materiali 
et,1,42 !!elementi piani di appoggio per la mesh solida 
et,2,45 
mp,ex,1,210000 
*end 
/prep7 
*use,ass_par 
c*** modellazione 
rectng,0,bc/2,0,dp/2 
rectng,bc/2,Lp,0,dp/2 
rectng,Lp,La/2,0,dp/2 
rectng,Lp,La/2,dp/2,da/2 
c*** mesh 
esize,3 
mshkey,1 



amesh,all !!mesh di superficie per l'estrusione 
type,2  
extopt,esize,10 !!parametri di estrusione: divisioni sull'arco di estrusione 
extopt,aclear,1 !!ripulisce le mesh di superficie di partenza 
vrotat,all,,,,,,1,2,90 !!mesh di estrusione 
*get,nmx,node,,num,max 
nsym,y,nmx,all 
esym,,nmx,all 
nummrg,node 
c*** vincoli e carichi 
nsel,s,loc,z,-0.001,0.001 !!vincoli di simmetria 
d,all,uz,0 
nsel,s,loc,x,La/2-0.001,La/2+0.001 
d,all,ux,0 
alls 
ncu=node(bc/2,0,0) !!cerniera sferica per cuscinetto 
d,ncu,all,0 
nf=node(La/2,da/2,0) !!carico tot:4000 N;  
f,nf,fy,-1000  !!su un quarto del modello: 1000 N 
finish 
/solu 
save !!salva il modello 
solve 
finish 
c*** creazione del sottomodello 
/clear 
/filname,sottomodello 
/prep7 
c*** si ridefiniscono elementi, materiale e parametri tramite la macro prima definita 
*use,ass_par 
dx=2*r !!distanza dalla discontinuità direzione x 
dy=2*r !!distanza dalla discontinuità direzione y 
c*** modellazione 
k,1,0,0 !!kp che definiscono l'asse di rotazione per l'estrusione 
k,2,10,0 
rectng,Lp-r,Lp+dx,-(dp/2+r),-(dp/2-dy) 
cyl4,Lp-r,-(dp/2+r),r !!raccordo 
asba,1,2 
rectng,Lp-dx,Lp-r,-(dp/2),-(dp/2-dy) 
rectng,Lp,Lp+dx,-(dp/2+2*r),-(dp/2+r) 
aadd,all !!si genera un'unica area 
lsel,s,loc,y,-(dp/2-dy)-0.001,-(dp/2-dy)+0.001 
lsel,a,loc,x,Lp+dx-0.001,Lp+dx+0.001 
lccat,all !!concatenazione di linee per mesh strutturata 
lsel,s,loc,y,-da/2,-dp/2+0.001 
lsel,r,loc,x,0,Lp+0.001 
lccat,all 
esize,0.5 
mshkey,1 
amesh,all 
lsel,s,lccat 
ldel,all 
alls 
type,2 
extopt,esize,12 !!parametri di estrusione 
extopt,aclear,1 
vrotat,all,,,,,,1,2,-30 
nummrg,node 
c*** identificazione nodi d'interfaccia tra i due modelli 
local,11,1,0,0,0,,,90 
asel,s,loc,x,dp/2,dp/2+2*r-0.001 
asel,r,loc,z,Lp-2*r,Lp 



cm,aa,area 
asel,r,loc,z,Lp-r,Lp-0.001 
cm,ab,area 
alls 
cmsel,u,aa 
nsla,,1 
nsel,r,loc,x,0,dp/2 
cm,na,node 
nsla,,1 
nsel,r,loc,z,Lp,Lp+dx 
cm,nb,node 
cmsel,s,na 
cmsel,a,nb  
c*** preparazione del file dei nodi 
nwrite 
alls 
save 
finish 
c*** interpolazione degli spostamenti 
resume,coarse,db 
/post1 
file,coarse,rst 
set,1,1 
cbdof !!effettua l'interpolazione per i nodi salvati nel file 
finish 
c*** si applicano gli spostamenti interpolati 
resume,sottomodello,db 
/solu 
/input,sottomodello,cbdo !!applica gli spostamenti interpolati al sottomodello 
solve 
finish 
c*** si visualizzano i risultati nel sottomodello 
/post1 
 
Nelle figure seguenti sono illustrate le caratteristiche generali dei modelli, ed il confronto fra le tensioni di 
interfaccia. 
 

 
 

Figg.2 e 3. Caratteristiche geometriche del modello coarse e del sottomodello. 
 

 
 



 
 
 

 
 

Figg.4 e 5. Caratteristiche della mesh del modello coarse e del sottomodello. 
 
 
 

 
 

Fig.6. Confronto fra le tensioni calcolate all’interfaccia. 
 


